LIGO Laboratory / LIGO Scientific Collaboration

LIGO-T060173-00-R | *LIGO* | Date August 3 2006

Measuring Coating Mechanical Quality Factors in a Layered Cantilever Geometry: a Fully Analytic Model

Vincenzo Pierro, Innocenzo M. Pinto

TWG, University of Sannio at Benevento, ITA

Distribution of this document:
LIGO Science Collaboration

This is an internal working note
of the LIGO Project.

California Institute of Technology
LIGO Project – MS 18-34
1200 E. California Blvd.
Pasadena, CA 91125
Phone (626) 395-2129
Fax (626) 304-9834
E-mail: info@ligo.caltech.edu

Massachusetts Institute of Technology
LIGO Project – NW17-161
175 Albany St
Cambridge, MA 02139
Phone (617) 253-4824
Fax (617) 253-7014
E-mail: info@ligo.mit.edu

LIGO Hanford Observatory
P.O. Box 1970
Mail Stop S9-02
Richland WA 99352
Phone 509-372-8106
Fax 509-372-8137

LIGO Livingston Observatory
P.O. Box 940
Livingston, LA 70754
Phone 225-686-3100
Fax 225-686-7189

http://www.ligo.caltech.edu/

1. Purpose and Motivation

Precise knowledge of the mechanical loss angles and Young moduli of the mirror coating constituents is needed for optimizing the coating multilayer thicknesses, so as to minimize the coating thermal noise for a prescribed reflectivity in advanced interferometric gravitational wave detectors [1].
In the simplest yet reliable model of the coating noise PSD, this latter is a linear combination of the total (physical) thicknesses $s_{H,L}$ of the coating low (suffix L) and high (suffix H) index constituents, currently, Tantala (Ta_2O_5) and Silica (SiO_2),

$$PSD = \Pi(s_L + \gamma s_H)\,. \tag{1}$$

In (1) Π is a constant of no concern to us here, and γ is basically given by the ratio $Y_H\phi_H / Y_L\phi_L$, where $Y_{H,L}$. and $\phi_{H,L}$ denote the Young moduli and mechanical loss-angles of the high and low index materials, respectively.
Presently available estimates for the loss angles $\phi_{H,L}$ are deduced from ringdown measurements based on the experimental setup and computational framework described in [2], using suspended thin or thick circular samples with different s_H/s_L ratios.
A number of related critical issues are noted: i) the ringdown modes are in the *KHz* range, and should be extrapolated down to the band of interest; ii) the retrieved loss angles depend critically on the value of the ratio between the elastic energies stored in the coating and substrate, which is computed numerically using finite elements (known to be of limited accuracy when dealing with thin layers); iii) the systematic error may be exceedingly large for materials with extremely low mechanical losses.
The $Y_{H,L}$ values, on the other hand, are presently approximated by the corresponding bulk-material values, although this may be a crude approximation[1].
Use of trustfully more accurate values for the Young modulus and the coating/substrate energy ratio led to a difference by factor two for the estimated Silica loss angle between [2] and [3], based on the same measurements. Available estimates for γ are thus presently affected by relatively large uncertainties[2].
Quite recently, an alternative experimental setup has been proposed, based on a clamped cantilever geometry. Potentially this may offer several advantages, including: i) easier/cheaper sample preparation and measurement setup (hopefully resulting in more abundant and more repeatable measurements); ii) ringdown modes (and derived quantities) falling directly within the frequency band of interest; iii) last but not least, the possibility of a *fully analytic* modelling, which is the subject of the present report.

Preliminary measurements based on the clamped cantilever are ongoing both at LMA, Lyon FR, and the University of Glasgow UK, yielding encouragingly consistent results [4].

[1] From a suitably large set of ringdown measurements, one may retrieve in principle both the loss angles and the Young moduli of the coating constituents. However, from ringdown measurements corresponding to different coating thicknesses, one may directly estimate the value of γ, without having to know separately the Young moduli and loss angles.

[2] Measurement-related errors may be expected to be Poisson-distributed; technology-related (coating process-repeatability related) uncertainties may be expected to be Gaussian-distributed. The number of presently available measurements is unfortunately so small to make the above distinctions of almost no practical use..

2. Clamped Cantilever Model

The basic model of a clamped coated cantilever in the small-oscillations regime has been developed in [5,6]. We refer henceforth to the geometry and notation in Figure 1.



Fig. 1 – Problem's geometry and notation

where the cantilever is assumed as being clamped at $y = 0$.
Define the neutral section $z = z_0$ by [7]:

$$\int_0^{h_s+h_c} (z - z_0) E(z) dz = 0, \qquad (2)$$

where $E(z)$ is the z-dependent Young modulus.
The layered cantilever is described (in the linearized homogeneization approach) by the equation

$$w(h_s \rho_s + h_c \rho_c) \frac{\partial^2 \zeta}{\partial t^2} = -w \frac{\partial^4 \zeta}{\partial y^4} \int_0^{h_s+h_c} (z - z_0)^2 E(z) dz, \qquad (3)$$

where ρ_s, ρ_c are the substrate and coating-layer mass-densities.
Introducing the complex (phasor) notation $\zeta(y,t) = \mathrm{Re}[z(y) e^{i\omega_n t}]$, the above equation becomes:

$$\frac{d^4 z(y)}{dy^4} = \kappa_n^4 z(y), \qquad (4)$$

where ω_n and κ_n are the angular frequency and wavenumber of the n-th mode, respectively. The dispersion equation connecting the above quantities reads:

$$\kappa_n^4 = \frac{\omega_n^2 (h_s \rho_s + h_c \rho_c)}{\int_0^{h_s+h_c} (z - z_0)^2 E(z) dz}. \qquad (5)$$

The natural modes are obtained by enforcing the boundary conditions:

$$z(0) = 0 \quad z'(0) = 0 \quad z''(L) = 0 \quad z'''(L) = 0. \qquad (6)$$

pertaining to the clamped and free end of the cantilever, respectively. The modes can be computed in explicit form, and are:

$$z(y) = A(\cos\kappa_n y - \cosh\kappa_n y) + B(\sin\kappa_n y - \sinh\kappa_n y)\ . \tag{7}$$

The A and B coefficients are defined but for a common multiplicative factor. Their ratio is:

$$\frac{A}{B} = \frac{\cos\kappa_n L - \cosh\kappa_n L}{\sin\kappa_n L - \sinh\kappa_n L}\ . \tag{8}$$

The product $\kappa_n L = x_n$ can be taken to represent the eigenvalue, and is obtained by solving the (trascendental) equation

$$\cos x_n \cosh x_n = -1. \tag{9}$$

3. Elastic Energy

The elastic energy per unit volume, assuming $\zeta = \zeta(y,t)$ is given by [7]

$$W = (z - z_0)^2 \frac{E(z)}{2\left(1-\sigma(z)^2\right)}\left(\frac{\partial^2\zeta}{\partial y^2}\right)^2 , \tag{10}$$

where $E(z)$ and $\sigma(z)$ are the z-dependent Young modulus and Poisson coefficient, and z_0 is the neutral section defined by (2). The average over a cycle of oscillation is expressed, in the phasor notation, by:

$$< W >= \frac{1}{4}(z - z_0)^2 \frac{E(z)}{\left(1-\sigma(z)^2\right)}\left|\frac{d^2 z}{dy^2}\right|^2 \ . \tag{11}$$

Upon integrating on the relevant volumes, the elastic energy in the coating (suffix c) and substrate (suffix s) of the cantilever can be written:

$$\begin{aligned} < W_s >_1 &= \frac{1}{4} w \int_0^{h_s} (z - z_0)^2 \frac{E(z)}{\left(1-\sigma(z)^2\right)} dz \int_0^L \left|\frac{d^2 z}{dy^2}\right|^2 dy \\ < W_c >_2 &= \frac{1}{4} w \int_{h_s}^{h_s+h_c} (z - z_0)^2 \frac{E(z)}{\left(1-\sigma(z)^2\right)} dz \int_0^L \left|\frac{d^2 z}{dy^2}\right|^2 dy \end{aligned} \ . \tag{12}$$

The ratio between the (one-cycle averaged) energies stored in the coating and substrate layers is accordingly:

$$\frac{< W_c >}{< W_s >} = \frac{\int_{h_s}^{h_s+h_c} (z-z_0)^2 \frac{E(z)}{\left(1-\sigma(z)^2\right)} dz}{\int_0^{h_s} (z-z_0)^2 \frac{E(z)}{\left(1-\sigma(z)^2\right)} dz} . \tag{13}$$

4. Ringdown and Coating Loss Angle

The coated cantilever quality factor in the flexural (lowest) eigenmode with resonant frequency ω_1 is:

$$Q = \frac{\omega_1 < W >}{< P >}, \tag{14}$$

where <*W*> and <*P*> are the elastic energy stored and the power dissipated averaged over a cycle [8]. On the other hand

$$Q_s = \frac{\omega_1 < W_s >}{< P_s >} \quad \text{and} \quad Q_c = \frac{\omega_1 < W_c >}{< P_c >} \quad , \tag{15}$$

where the suffixes c and s refer to the coating and substrate as usual. Hence:

$$Q_c^{-1} = Q^{-1} + \frac{< W_s >}{< W_c >}\left(Q^{-1} - Q_s^{-1}\right) . \tag{16}$$

The ratio in (16) can be evaluated using (13). A limiting form of the averaged energy ratio in (13) valid under the assumptions $<W_c> << <W_s>$ has been given in [9,10] and is:

$$\frac{< W_s >}{< W_c >} \approx \frac{E_s h_s}{3E_c h_c} \quad , \tag{17}$$

where E_s and E_c are the Young moduli of the substrate and coating, respectively. A MATHEMATICA™ notebook implementing the above formulas has been written and is made available in the Appendix. Note that the *uncoated* cantilever quality factor differs from the substrate quality factor in (16) by less then 1%. Equation (16) allows to determine the coating loss angle (inverse of coating quality factor) from measured damping constants of the bare (uncoated) and coated cantilever.
Extension of the above fully analytic model to (heterogeneous) multilayer-coated cantilevers is also possible [11] and relatively straightforward, and will be the subject of a forthcoming report.

References

[1] J. Agresti et al., LIGO G050363-00-R; LIGO P060027-00-Z.

[2] S.D. Penn et al., Class. Quantum Grav., **20** (2003) 2917.

[3] D.R.M. Crooks et al., Class. Quantum Grav., **23** (2006) in print.

[4] N. Morgado and S. Rowan, personal communication (2006).

[5] R. Sandberg et al., J. Micromech. Microeng., **15** (2005) 1454.

[6] R. Sandberg et al., J. Micromech. Microeng., **15** (2005) 2249.

[7] L.D. Landau and E.M. Lifschitz, , *Theory of Elasticity* (1987) Pergamon Press, New York.

[8] Z.S. Li et al., Material Sci. Eng., **A370** (2004) 186.

[9] Y. Nishino and S. Asano, Phys. Stat. Sol., **A139** (1993) K97.

[10] Y. Nishino and S. Asano, J. de Phys., **IV** (1996) 783.

[11] S. Vengallatore, J. Micromech. Microeng., **15** (2005) 2398.

APPENDIX – *A MATHEMATICA™ NOTEBOOK*

```
(***************************************************************************)
(*          Analytic model of clamped coated cantilever ringdown          *)
(*          MATHEMATICA 5.0 Notebook ver. 1, 19/04/2006                    *)
(*          Authors: V. Pierro, I.M Pinto, TWG University of Sannio, INFN and LSC   *)
(***************************************************************************)

(* Spelling warnings turned off *)
Off[General::spell];
Off[General::spell1];

(* thickness [m] *)
hs=120*10^-6  (* substrate, Silica *);
hc=1/2*10^-6 (* coating, Tantala *);

(* length & width [m] *)
ll=40*10^-3;
w=5*10^-3;

(* Young moduli [Pa]*)
Es=70*10^9 (* substrate, Silica *);
Ec=140*10^9      (* coating, Tantala *);

(* densities [Kg/m^3]*)
rhos=2202 (* substrate, Silica *);
rhoc=8316 (* coating, Tantala *);

(* Poisson coefficients [] *)
sigs=0.16395 (* substrate, Silica *);
sigc =0.23 (* coating, Tantala *);

(* Piecewise-constant, z-dependent Young modulus [Pa] *)
EE[z_]:=Es*UnitStep[hs-z]+Ec*(UnitStep[z-hs]-UnitStep[z-hs-hc])

(* Neutral surface *)
zz0=z0/.Solve[Integrate[(z-z0)*EE[z],{z,0,hs+hc}]==0,z0][[1]]//N[#,22]&;
Print["Neutral surface position [μm] :",zz0*10^6]
```

```
(* Eigenvalues *)
xx[k_]:=x/.FindRoot[Cos[x]*Cosh[x]==-1,{x,Pi*(k-1/2)},WorkingPrecision→20]
ASB[k_]:=(Cos[xx[k]]+Cosh[xx[k]])/(Sin[xx[k]]-Sinh[xx[k]])

(* Eigenvalue list *)
Print["Eigenvalues: ", (xtav=Table[{k,xx[k]},{k,1,10}])//TableForm]

(* Resonant frequency, fundamental mode [Hz] *)

(* coated cantilever  *)
f1coated=(xx[1]/ll)^2*Sqrt[Integrate[(z-zz0)^2*EE[z],{z,0,hs+hc}]/(hs*rhos+hc*rhoc)]/(2*Pi);
(* substrate only; 0-thickness coating *)
f1uncoated=(xx[1]/ll)^2*Sqrt[Integrate[(z-hs/2)^2*Es,{z,0,hs}]/(hs*rhos)]/(2*Pi);

Print["f1, coated cantilever [Hz] : " , f1coated]
Print["f1, substrate only [Hz] : " , f1uncoated]
Print["f1coated/f1uncoated   :", f1coated/f1uncoated]

(* Dominant wavenumber [m^-1] *)
k1coated=Sqrt[2*Pi*f1coated]*(Integrate[(z-zz0)^2*EE[z],{z,0,hs+hc}]/(hs*rhos+hc*rhoc))^(1/4);

Print["Dominant wavenumber [m^-1] :",k1coated]

(* Elastic energy *)
Ws=w/(4*(1-sigs^2))*Integrate[(z-zz0)^2*EE[z],{z,0,hs}]*
     Integrate[D[zz1[y],{y,2}]^2,{y,0,ll}];
Wc=w/(4*(1-sigc^2))*Integrate[(z-zz0)^2*EE[z],{z,hs,hc+hs}]*
     Integrate[D[zz1[y],{y,2}]^2,{y,0,ll}];

(* Elastic energy, substrate only *)
Ws0=w/(4*(1-sigs^2))*Integrate[(z-
1/2*hs)^2*EE[z],{z,0,hs}]*Integrate[D[zz1[y],{y,2}]^2,{y,0,ll}];

Print["Ws0/Ws, full analytic : ", Ws0/Ws]

 (* Longitudinal distribution, dominant (flexural) mode *)
zz1[y_]=ASB[1]*(Cos[k1coated*y]-Cosh[k1coated*y] )+(Sin[k1coated*y]-Sinh[k1coated*y]);

(* Plot longitudinal distribution, scaled *)
Plot[zz1[y*ll]/zz1[ll],{y,0,1},Axes→False,Frame→True,AspectRatio→.3,
  FrameLabel→{"z/L",None},PlotLabel->"Normalized Mode-Amplitude",ImageSize→500]

(* Ratio Ws/Wc *)
rat=Ws/Wc;
(* Approximate value of ratio, according to Nishino, eq. (15) *)
ratapp= (Es*hs)/(3.*Ec*hc);

Print["Ratio Ws/Wc, exact, analytic : ", rat]
Print["Ratio Ws/Wc, Nishino's app. : ", ratapp]
```